


74HC595 8 Bit Schieberegister mit Latch (zu Aufgabenblatt 5)



PIN ASSIGNMENT			
Q _B	1	16	V _{CC}
Q _C	2	15	Q _A
Q _D	3	14	A
Q _E	4	13	OUTPUT ENABLE
Q _F	5	12	LATCH CLOCK
Q _G	6	11	SHIFT CLOCK
Q _H	7	10	RESET
GND	8	9	SQ _H

Output Enable: Schaltet die Ausgangspins Q_A bis Q_H ein (=LOW) oder aus (=HIGH). Dieser Pin sollte also normalerweise mit GND verbunden werden.

Latch Clock: Die im Schieberegister gespeicherten

Werte werden mit „HIGH“ in das Storage Register übernommen. Dieses Ausgaberegister wird meistens mit **Latch** bezeichnet.

A: Hier stehen die **Daten** (DS) an, die in das Schieberegister übernommen werden sollen.

Shift Clock: Die an Pin A anliegenden Daten werden bei einem Signalwechsel von L auf H in das Schieberegister übernommen.

Reset: Wie der Name schon vermuten lässt wird hier der Inhalt des Schieberegisters mit „LOW“ (=GND) zurückgesetzt. Pin 10 muss also normalerweise mit +5V verbunden werden.

SQ_H: Das tolle an Schieberegistern ist, dass man mehrere davon hintereinander, also in Reihe, schalten kann. Ist das Schieberegister voll so kann man die Daten die sonst sozusagen hinten wieder herausfallen würden in das nächste Schieberegister übergeben.

Sketch 1: (denkbar einfachste Anwendung ohne Schnick-Schnack ☺)

Wir übernehmen bei jedem Bit der Dateneingabe den Inhalt des Shift-Registers in den Speicher (Latch), damit man den Vorgang beobachten kann. Vorher wird das Shift-Register noch gelöscht.

```
int shiftPin = 8; //entspricht „clockPin“ (=Shift Clock), bei L>H erfolgt Übernahme ins Shift-Register
int storePin = 9; // entspricht „latchPin“ (=Latch Clock), bei L>H erfolgt Übernahme in den Speicher
int dataPin = 10; // die Daten müssen anliegen, bevor die Shift-Clock von L>H geht
int resetPin = 11; // zum Rücksetzen des Schieberegisters
int muster[8] = {1,1,0,1,0,0,0,1}; // Dieses Muster soll z.B. ausgegeben werden

void setup() {
  pinMode(storePin, OUTPUT);
  pinMode(shiftPin, OUTPUT);
  pinMode(dataPin, OUTPUT);
  pinMode(resetPin, OUTPUT);

  digitalWrite(resetPin, LOW); // die nächsten 4 Anweisungen setzen das Schieberegister
  digitalWrite(storePin, LOW); // und das Display (LEDs) zurück
  digitalWrite(storePin, HIGH);
  digitalWrite(resetPin, HIGH);
}
```

```

for (int i=0; i<8; i++) {           // Array auslesen

delay(1000);                        // damit der Vorgang langsam abläuft
digitalWrite(storePin, LOW);        // storePin wieder auf LOW, Aktion passiert nur bei Wechsel von L auf H
digitalWrite(shiftPin, LOW);        // wie oben
digitalWrite(dataPin, muster[i]);   // Jetzt den Wert der aktuellen Stelle ans Datenpin „A“ (DS) anlegen
digitalWrite(shiftPin, HIGH);        // Übernahme ins Shift-Register
digitalWrite(storePin, HIGH);        // Übernahme in den Speicher
}                                    // nach 8 Umläufen: Ende der Schleife
}                                    // Ende Setup

void loop () {
  // Hier passiert nichts.
}

```

Sketch 2:

Im Arduino Sketch 2 wird ein Zähler von 0 bis 255 gezählt und der jeweilige Wert in binärer Darstellung auf den 8 LEDs ausgegeben. Zur Übertragung des 8-bit-Wertes in das Schieberegister wird hier der [shiftOut\(\)](#)-Befehl des Arduino verwendet. Dieser erledigt die Übertragung der einzelnen Bits ins Schieberegister inklusive der notwendigen Flankenwechsel am Shift Pin.

ShiftOut (); Verschiebt ein Daten-Byte ein Bit nach dem anderen. Startet entweder aus dem größten (d. h. dem am weitesten links liegenden) oder aus dem am wenigsten (rechtsten) signifikanten Bit. Jedes Bit wird wiederum auf einen Dateneingang geschrieben, wonach ein Takt gepulst wird (HIGH und dann LOW). Hinweis: Man muss sicherstellen, dass der Takt vor dem Aufruf von shiftOut () „LOW“ ist. Das geht mit der Anweisung digitalWrite (clockPin, LOW).

Syntax: ShiftOut (dataPin, clockPin, bitOrder, value)

bitOrder: Reihenfolge um die Bits zu verschieben; Entweder MSBFIRST oder LSBFIRST.

(Das **m**eiste signifikante **B**it zuerst [**f**irst]) oder das am wenigsten [**l**etzte] signifikante **B**it zuerst

value: Wert der zu übertragenden Daten, Angabe als Dezimalzahl oder mit **0b.....** als Binärzahl)

Die Übertragung des Schieberegister-Inhalts an das Latch mit einem Wechsel des Latch-Clock Pin (=storePin) von LOW zu HIGH müssen wir dann noch selbst erledigen.

```

int shiftPin = 8;
int storePin = 9;
int dataPin = 10;
int counter = 0;

counter ++;
if (counter > 255) {
  counter = 0;
}
}

```

```

void setup() {
  pinMode(storePin, OUTPUT);
  pinMode(shiftPin, OUTPUT);
  pinMode(dataPin, OUTPUT);
}

```

```

void loop () {

  digitalWrite(storePin, LOW);
  shiftOut(dataPin, shiftPin, MSBFIRST, counter);
  digitalWrite(storePin, HIGH);
  delay(500);
}

```

Binärer Zähler von 0 bis 255.

1 = **0b**000000001
 2 = **0b**000000010
 3 = **0b**000000011
 4 = **0b**000000100

.....

usw. Die LEDs zeigen alle Werte von 0 bis 255 an.

Sketch 3:

Neue Anweisungen: **bitSet(x,n)**; schreibt eine „1“ in einem Bit einer (numerischen) Variablen.

x: Variable, in der ein Bit gesetzt werden soll (hier „leds“), n: das zu setzende Bit, beginnend bei 0 für das niedrigstwertige Bit (ganz rechts), siehe auch „Erläuterungen“.

Dieser Sketch bewirkt, dass ein Schieberegister „langsam“ gefüllt wird und damit immer mehr LEDs leuchten.

Nebenbei: Mit **bitClear(x,n)**; könnte man einzelne Bits wieder löschen (auf 0 setzen).

<https://learn.adafruit.com/adafruit-arduino-lesson-4-eight-leds/arduino-code>

```
int clockPin = 8;
int latchPin = 9;
int dataPin = 10;

byte leds = 0;

void setup()
{
  pinMode(latchPin, OUTPUT);
  pinMode(dataPin, OUTPUT);
  pinMode(clockPin, OUTPUT);
}

void loop()
{
  leds = 0;
  updateShiftRegister(); // Methode *)
  delay(500);

  for (int n = 0; n < 8; n++)
  {
    bitSet(leds, n);
    updateShiftRegister(); //Methode *)
    delay(500);
  }
}

void updateShiftRegister() // *)
{
  digitalWrite(latchPin, LOW);
  shiftOut(dataPin, clockPin, LSBFIRST, leds);
  digitalWrite(latchPin, HIGH);
}
```

Hier ein paar Erläuterungen zum Sketch:

Etwas gewöhnungsbedürftig ist, dass einzelne Bits betrachtet und verschoben werden. Egal in welcher Form wir Zahlen eingeben, unser Rechner wandelt alles in Binärzahlen um.

Wenn die Variable leds auf Null gesetzt wird, bedeutet das, dass $0_{10} = 00000000_2$ ist, also die dezimale Null dem binären Byte 00000000 entspricht. Wenn kein Index dabei steht ist die Zahl immer im Zehnersystem zu verstehen. Soll es eine Binärzahl sein, müsste man z.B. schreiben: 130 = **0b**10000010. Man notiert 0b vor der Zahl, damit es hier nicht mit 10-Millionen-10 verwechselt werden kann.

Die nebenstehenden Anweisungen bewirken, dass die Variable **leds**, die verändert werden soll, zunächst auf Null gesetzt wird. Das Laden und Anzeigen im Shift-Register wird durch eine **Methode** **updateShiftRegister()** vorgenommen, die nach unten „ausgelagert“ wurde. Das ist hier praktisch, da sie zweimal verwendet wird.

Die Anweisung **bitSet** bewirkt, dass in der Schleife nacheinander jeweils ein zusätzliches Bit gesetzt wird. Wenn also die Variable leds vorher den Wert leds= 7 hatte (also **0b**00000111), dann wird sie nach der Anweisung **bitSet(leds, 4)** zu **0b**0000**1**111, das 4. Bit wurde zusätzlich gesetzt. Nebenbei hat die Variable jetzt den Wert 15.

Im Loop wird also erst das Register gelöscht, so dass alle LEDs aus sind und dann werden durch die Schleife die einzelnen Bits gesetzt und angezeigt.

Man könnte die ausgelagerte Methode z.B. auch mit „schreibeAufSchiebeRegister();“ bezeichnen. Dann hätte man den Sketch aber zweisprachig. Das sollte vermeiden werden.

Sketch 4: Brightness Control

<https://learn.adafruit.com/adafruit-arduino-lesson-4-eight-leds/brightness-control>

One pin of the 74HC595 that I have not mentioned is a pin called 'Output Enable'. This is pin 13 and on the breadboard, it is permanently connected to Ground. This pin acts as a switch, that can enable or disable the outputs - the only thing to watch for is it is 'active low' (connect to ground to enable).

So, if it is connected to 5V, all the outputs go off.

We can use this pin along with the **analogWrite** function, to control the brightness of the LEDs using PWM.

To do this, all you need to do, is to change the connection to pin 13 of the 74HC595 so that instead of connecting it to Ground, you connect it to pin 3 of the Arduino.

The sketch 3 will once all the LEDs have been lit gradually fade them back to off.

```
int clockPin = 8;
int latchPin = 9;
int dataPin = 10;
int outputEnablePin = 3;

byte leds = 0;

void setup()
{
  pinMode(latchPin, OUTPUT);
  pinMode(dataPin, OUTPUT);
  pinMode(clockPin, OUTPUT);
  pinMode(outputEnablePin, OUTPUT);
}

void loop()
{
  setBrightness(255);    // ausgelagerte Methode
  leds = 0;
  updateShiftRegister(); // ausgelagerte Methode
  delay(500);
  for (int i = 0; i < 8; i++)
  {
    bitSet(leds, i);
    updateShiftRegister();
    delay(500);
  }
  for (byte b = 255; b > 0; b--)
  {
    setBrightness(b);
    delay(50);
  }
}

void updateShiftRegister()
{
  digitalWrite(latchPin, LOW);
  shiftOut(dataPin, clockPin, LSBFIRST, leds);
  digitalWrite(latchPin, HIGH);
}

void setBrightness(byte brightness)    // 0 bis 255
{
  analogWrite(outputEnablePin, 255-brightness);
}
```